



How To Find Killing Vectors

Synopsis

- Let g be a (pseudo) Riemannian metric on a manifold M . Let ∇_a be the [covariant derivative](#) defined by the [Christoffel connection](#) of the metric g , and let $K_a = g_{ab}K^b$ be the dual vector corresponding to the vector field K^a . Then K^a is a Killing vector field if and only if it solves the Killing equations:

$$\nabla_a K_b + \nabla_b K_a = L_K g_{ab} = 0.$$

- The [flow](#) of a Killing vector field is a 1-parameter family of isometries of (M, g) . The set of Killing vectors of a given metric form a Lie algebra using the commutator of vector fields as the [Lie bracket](#).
- In this worksheet we show how to compute the Lie algebra of Killing vector fields of a metric using the commands [KillingVectors](#) and [LieAlgebraData](#).

Example

Here we compute the Lie algebra of Killing vector fields of the Gödel metric, a metric featuring in the general theory of relativity.

Load in the required packages. Set typesetting [preferences](#) (optional).

```
[> with(DifferentialGeometry): with(Tensor): with(LieAlgebras):  
[> Preferences("TensorDisplay", 1):
```

Define a coordinate chart.

```
[> DGsetup([t, x, y, z], M);  
frame name: M (1)
```

Here is the Gödel metric.

```
[> g := evalDG(dx &t dx + dy &t dy + 1/2*exp(2*x)*dz &t dz - (dt + exp  
(x)*dz) &tensor (dt + exp(x)*dz));  
g := -dt ⊗ dt - ex dt ⊗ dz + dx ⊗ dx + dy ⊗ dy - ex dz ⊗ dt -  $\frac{e^{2x}}{2}$  dz ⊗ dz (2)
```

The command [KillingVectors](#) computes a basis for the vector space of solutions to the Killing equations.

```
[ > KV := KillingVectors(g);
  KV := [-4 e^{-x} D_t + 2 z D_x - (z^2 - 2 e^{-2x}) D_z, 2 D_x - 2 z D_z, -2 D_z, -D_t, D_y] (3)
```

We see that the Gödel metric admits a 5-dimensional vectors space of Killing vectors.

```
[ M > nops(KV);
  5 (4)
```

We can verify that the vector fields in the list KV are Killing vectors with the [LieDerivative](#) command.

```
[ M > LieDerivative(KV, g);
  [0 dt ⊗ dt, 0 dt ⊗ dt, 0 dt ⊗ dt, 0 dt ⊗ dt, 0 dt ⊗ dt] (5)
```

We can also verify that the Killing vectors form a Lie algebra using the command [LieAlgebraData](#). The algebra elements corresponding to the vector fields are labeled e_1, e_2, \dots, e_5 . Only the non-zero brackets are displayed.

```
[ M > LAD := LieAlgebraData(KV);
  LAD := [[e1, e2] = 2 e1, [e1, e3] = 2 e2, [e2, e3] = 2 e3] (6)
```

The Lie algebra defined by LAD can be analyzed using commands in the [LieAlgebras](#) package.

Commands Illustrated

- [DGsetup](#), [evalDG](#), [KillingVectors](#), [LieAlgebraData](#), [LieDerivative](#)

Related Commands

- [KillingTensors](#), [KillingSpinors](#)

References

- http://en.wikipedia.org/wiki/Killing_vector_field
- Choquet-Bruhat, Yvonne, DeWitt-Morette, Cécile, *Analysis, Manifolds and Physics*, (Amsterdam: Elsevier, 1977).

Release Notes

- The illustrated commands are available in Maple 13 and subsequent releases.

Authors

Charles Torre
 Department of Physics, Utah State University
 February 22, 2013

